```
                                    ;FADA2D3.DOC
                ;            ** TIC3 I/P AND A/D CONVERTER FOR FADE RATE **
                ;*****************************************************************
                ;* A MULTI-PORT,QUADRATURE-PHASE THYRISTOR CONTROLLER. WRITTEN IN    *
                ;* 68HC11 ASSEMBLY LANGUAGE USING CROSS-32 META-ASSEMBLER,EDITED WITH*
                ;* MULTI-EDIT ON A DATACORP 286 MS-DOS COMPUTER.                     *
                ;* THIS PROGRAM, WITH APPROPRIATE HARDWARE WILL SIMULTANEOUSLY FADE  *
                ;* IN AND FADE OUT FOUR 120VAC LOADS, 90 DEG OUT OF PHASE WITH EACH  *
                ;* OTHER. IT USES THE INPUT CAPTURE AND OUTPUT COMPARE FEATURES OF    *
                ;* THE MC68HC11 MCU, THE A/D CONVERTER, AND REAL TIME INTERRUPTS.    *
                ;*****************************************************************
                ;*                         FOUND ENGINEERING 1992                *
                ;*****************************************************************

 0000                   CPU     "68HC11.TBL"
 0000                   HOF     "MOT8"
 B600                   ORG     $B600     ;START OF EEPROM IN 68HC11
 B600                   TITL    "** FADER  Version 1.50 **"
 B600                   page    64
 B600                   page
```



```
                ;*****************************************************************
                ;*                                EQUATES                        *
                ;*****************************************************************

 1000 =         REGBAS:  EQU 1000H        ;START ADDR OF REGISTER BLOCK
 0004 =         PORTB:   EQU 04H          ;OUTPUT PORT B
 000E =         TCNT:    EQU 0EH          ;FREE RUNNING COUNTER
 0014 =         TIC3:    EQU 14H          ;INPUT CAPTURE REGISTER
 0018 =         TOC2:    EQU 18H          ;OUTPUT COMPARE REGISTER 2
 001A =         TOC3:    EQU 1AH          ;TOC1 IS NOT USED
 001C =         TOC4:    EQU 1CH
 001E =         TOC5:    EQU 1EH          ;ALSO USED BY TRACE IN EVBU
 0020 =         TCTL1:   EQU 20H          ;OUTPUT MODE ON COMPARE REG
 0021 =         TCTL2:   EQU 21H          ;EDGE DETECT CONFIGURATION
```

```
0022 =          TMSK1:   EQU 22H          ;MASK REG TO ENABLE INTERRUPTS
0023 =          TFLG1:   EQU 23H          ;INTERRUPT FLAG REGISTER
0030 =          ADCTL:   EQU 30H          ;A/D CONTROL/STATUS
0032 =          FADRATE: EQU 32H          ;A/D RESULT REGISTER=FADE RATE
0039 =          OPTION:  EQU 39H          ;CONTROLS A/D POWER-UP
00DC =          PVOC2:   EQU 00DCH        ;PSEUDO VECTOR FOR OUTPUT CAPTURE ISR
00D9 =          PVOC3:   EQU 00D9H
00D6 =          PVOC4:   EQU 00D6H
00D3 =          PVOC5:   EQU 00D3H
0001 =          OUT1:    EQU 00000001B    ;PORT B, PIN 1
0002 =          OUT2:    EQU 00000010B
0004 =          OUT3:    EQU 00000100B
0008 =          OUT4:    EQU 00001000B
0001 =          OUT1CLR: EQU OUT1         ;PORT PIN CLEAR MASK (0UT=LOW)
0002 =          OUT2CLR: EQU OUT2
0004 =          OUT3CLR: EQU OUT3
0008 =          OUT4CLR: EQU OUT4
00BF =          FLG1CLR: EQU 10111111B    ;CLEAR LOCAL INTERRUPT FLAG MASK
00DF =          FLG2CLR: EQU 11011111B
00EF =          FLG3CLR: EQU 11101111B
00F7 =          FLG4CLR: EQU 11110111B
007E =          JUMPCOD: EQU 07EH
01FF =          STACK:   EQU 01FFH        ;TOP OF USER RAM
0003 =          EDGEDET: EQU 00000011B    ;DETECT RISING AND FALLING EDGES
00FF =          SETOUT:  EQU 11111111B    ;SET O/P ON SUCCESSFUL COMPARE
0079 =          CLRFLAG: EQU 01111001B    ;CLEAR IC3/OC2,3,4,5 FLAGS
0078 =          INTMASK: EQU 01111000B    ;ENABLE OCx, MASK IC3 FOR POLL
00FE =          CLRIC3:  EQU 11111110B    ;CLEAR IC3 FLAG
0001 =          CROSS:   EQU 00000001B    ;IC3 FLAG DETECT
0080 =          ENABA2D: EQU 10000000B    ;START A/D
0021 =          CONVERT: EQU 00100001B    ;CONT SCAN,SINGLE CHANNEL, CH 1
                ;REG Y=USED FOR DELAY VALUES
                ;REG X=HOLDS REGISTER BLOCK ADDRESS
                ;REG D=GENERAL PURPOSE 16 BIT REG
B600                     page


** FADER  Version 1.50 **    PAGE 3    Sun Oct 11 14:40:09 1992

                ;*****************************************************************
                ;*                          INITIALIZATION                       *
                ;*****************************************************************
```

```
B600 0F        GENINIT:  SEI                ;DISABLE INTERRUPTS DURING INIT
                ;*****INITIALIZE DELAY VALUES FOR ALL PORTS*****
B601 18FEB7AB             LDY     INIT1
B605 18FF0100             STY     DELAY1
B609 18FEB7AD             LDY     INIT2
B60D 18FF0102             STY     DELAY2
B611 18FEB7AF             LDY     INIT3
B615 18FF0104             STY     DELAY3
B619 18FEB7B1             LDY     INIT4
B61D 18FF0106             STY     DELAY4
                ;*****INITIALIZE FADE-IN/FADE-OUT DIRECTION FLAGS*****
B621 B6B7A9               LDAA    DECFLAG     ;FADE IN FLAG (=00)
B624 B70108               STAA    FLAG1       ;DECREASE DELAY,INC CONDUCTION
B627 B70109               STAA    FLAG2
B62A B6B7AA               LDAA    INCFLAG     ;FADE OUT FLAG
B62D B7010A               STAA    FLAG3       ;INCREASE DELAY,DEC CONDUCTION
B630 B7010B               STAA    FLAG4
B633 8E01FF               LDS     #STACK      ;TOP OF USER STACK
                ;*****SET UP INTERRUPT VECTORS*****
B636 867E                 LDAA    #JUMPCOD    ;JUMP EXTENDED OPCODE
B638 97DC                 STAA    PVOC2       ;OC2 PSEUDO VECTOR START ADDR
B63A 97D9                 STAA    PVOC3
B63C 97D6                 STAA    PVOC4
B63E 97D3                 STAA    PVOC5
B640 CCB760               LDD     #OUTSERV1
B643 DDDD                 STD     PVOC2+1     ;FINISH JUMP INST TO OCx ROUTINES
B645 CCB771               LDD     #OUTSERV2
B648 DDDA                 STD     PVOC3+1
B64A CCB782               LDD     #OUTSERV3
B64D DDD7                 STD     PVOC4+1
B64F CCB793               LDD     #OUTSERV4
B652 DDD4                 STD     PVOC5+1
                ;*****CONFIGURE REGISTERS AND PORTS*****
B654 CE1000               LDX     #REGBAS
B657 6F04                 CLR     PORTB,X
B659 8603                 LDAA    #EDGEDET    ;INPUT EDGE CONFIGURATION
B65B A721                 STAA    TCTL2,X     ;
B65D 8600                 LDAA    #00000000B  ;DISABLE TIMER OUTPUTS
B65F A720                 STAA    TCTL1,X
B661 8679                 LDAA    #CLRFLAG    ;INPUT CAPTURE/OUTPUT COMPARE FLAGS
B663 A723                 STAA    TFLG1,X
```

```
 B665 1C2278             BSET    TMSK1,X,INTMASK  ;ENABLE OCx, MASK IC3 FOR POLL
                ;*****START A/D CONVERTER*****
 B668 1C3980             BSET    OPTION,X,ENABA2D ;ENABLE A/D CONVERTER
 B66B 8621      LODRATE: LDAA    #CONVERT
 B66D A730               STAA    ADCTL,X          ;WRITE TO CONTROL TO START CONVERT
 B66F                    page


** FADER  Version 1.50 **    PAGE 4    Sun Oct 11 14:40:09 1992

                ;*********************************************************************
                ;*                              MAIN PROGRAM                         *
                ;*********************************************************************

 B66F 0E        START:   CLI                ;GLOBAL MASK: ALLOW ALL INTERRUPTS
 B670 1D23FE    CLEAR:   BCLR    TFLG1,X,CLRIC3   ;CLEAR IC3 FLAG ON EACH PASS
 B673 1F2301FC  POLL:    BRCLR   TFLG1,X,CROSS,POLL  ;WAIT FOR ZERO CROSS
                ;*****TIMING SECTION, WHERE ALL THE WORK GETS DONE*****
 B677 EC14      INPUT:   LDD     TIC3,X      ;GET TIMER COUNT FROM INPUT CAPT. REG
 B679 F30100             ADDD    DELAY1      ;COUNT VALUE AFTER DELAY
 B67C ED18               STD     TOC2,X      ;SET OUTPUT TO TRIGGER THEN
 B67E EC14               LDD     TIC3,X
 B680 F30102             ADDD    DELAY2
 B683 ED1A               STD     TOC3,X
 B685 EC14               LDD     TIC3,X
 B687 F30104             ADDD    DELAY3
 B68A ED1C               STD     TOC4,X
 B68C EC14               LDD     TIC3,X
 B68E F30106             ADDD    DELAY4
 B691 ED1E               STD     TOC5,X

                ;****************************PORT 1 SERVICE**************************

 B693 A632      START1:  LDAA    FADRATE,X     ;COUNT DOWN FASTER BY 1/N
 B695 18FE0100           LDY     DELAY1      ;GET VARIABLE DELAY VALUE
 B699 7D0108             TST     FLAG1       ;DELAY INCREASING OR DECREASING?
 B69C 2702               BEQ     DOWN1       ;00 = DECREASING, LAMP -> BRIGHT
 B69E 2016               BRA     UP1         ;FF = INCREASING, LAMP -> DIM
 B6A0 18BCB7A7  DOWN1:   CPY     MIN         ;SMALLEST DELAY VALUE?
 B6A4 270B               BEQ     TOGGLE1     ;YES, FLIP THE FLAG
 B6A6 1809               DEY                 ;NO, KEEP DECREASING
 B6A8 4A                 DECA                ;COUNT/N?
```

```
B6A9 26F5                BNE     DOWN1       ;NO, GO BACK FOR ANOTHER ROUND
B6AB 18FF0100            STY     DELAY1      ;UPDATE DELAY VALUE
B6AF 2014                BRA     START2      ;SERVICE NEXT PORT
B6B1 730108    TOGGLE1:  COM     FLAG1       ;COMPLEMENT FLAG, SO COUNT CAN
B6B4 200F                BRA     START2      ;CONTINUE IN OTHER DIRECTION
B6B6 18BCB7A5  UP1:      CPY     MAX         ;HIGHEST DELAY VALUE?
B6BA 27F5                BEQ     TOGGLE1     ;YES, FLIP THE FLAG
B6BC 1808                INY                 ;NO, KEEP INCREASING
B6BE 4A                  DECA
B6BF 26F5                BNE     UP1
B6C1 18FF0100            STY     DELAY1      ;UPDATE DELAY VALUE

               ;*****************************PORT 2 SERVICE*****************************

B6C5 A632      START2:   LDAA    FADRATE,X
B6C7 18FE0102            LDY     DELAY2
B6CB 7D0109              TST     FLAG2
B6CE 2702                BEQ     DOWN2
B6D0 2016                BRA     UP2
B6D2 18BCB7A7  DOWN2:    CPY     MIN
B6D6 270B                BEQ     TOGGLE2
B6D8 1809                DEY
B6DA 4A                  DECA
B6DB 26F5                BNE     DOWN2
B6DD 18FF0102            STY     DELAY2
B6E1 2014                BRA     START3
B6E3 730109    TOGGLE2:  COM     FLAG2
B6E6 200F                BRA     START3
B6E8 18BCB7A5  UP2:      CPY     MAX


** FADER  Version 1.50 **    PAGE 5    Sun Oct 11 14:40:09 1992

B6EC 27F5                BEQ     TOGGLE2
B6EE 1808                INY
B6F0 4A                  DECA
B6F1 26F5                BNE     UP2
B6F3 18FF0102            STY     DELAY2

               ;*****************************PORT 3 SERVICE*****************************

B6F7 A632      START3:   LDAA    FADRATE,X
```

```
B6F9 18FE0104            LDY     DELAY3
B6FD 7D010A              TST     FLAG3
B700 2702                BEQ     DOWN3
B702 2016                BRA     UP3
B704 18BCB7A7  DOWN3:    CPY     MIN
B708 270B                BEQ     TOGGLE3
B70A 1809                DEY
B70C 4A                  DECA
B70D 26F5                BNE     DOWN3
B70F 18FF0104            STY     DELAY3
B713 2014                BRA     START4
B715 73010A    TOGGLE3:  COM     FLAG3
B718 200F                BRA     START4
B71A 18BCB7A5  UP3:      CPY     MAX
B71E 27F5                BEQ     TOGGLE3
B720 1808                INY
B722 4A                  DECA
B723 26F5                BNE     UP3
B725 18FF0104            STY     DELAY3

               ;*****************************PORT 4 SERVICE*****************************

B729 A632      START4:   LDAA    FADRATE,X
B72B 18FE0106            LDY     DELAY4
B72F 7D010B              TST     FLAG4
B732 2702                BEQ     DOWN4
B734 2018                BRA     UP4
B736 18BCB7A7  DOWN4:    CPY     MIN
B73A 270C                BEQ     TOGGLE4
B73C 1809                DEY
B73E 4A                  DECA
B73F 26F5                BNE     DOWN4
B741 18FF0106            STY     DELAY4
B745 7EB670              JMP     CLEAR       ;START OVER, WAIT FOR NEXT ZERO
B748 73010B    TOGGLE4:  COM     FLAG4
B74B 7EB670              JMP     CLEAR
B74E 18BCB7A5  UP4:      CPY     MAX
B752 27F4                BEQ     TOGGLE4
B754 1808                INY
B756 4A                  DECA
B757 26F5                BNE     UP4
B759 18FF0106            STY     DELAY4
```

```
 B75D 7EB670              JMP     CLEAR       ;ALL 4 PORTS HAVE BEEN SERVICED IN
                                              ;FIRST 1/2 CYCLE, START OVER
 B760                     page
```



```
                ;*********************************************************************
                ;*                          INTERRUPT SERVICE ROUTINES                 *
                ;*********************************************************************

                ;*********************************************************************
                ;* THE OUTSERV INTERRUPT SERVICE ROUTINES ARE ENTERED WHEN THE DELAY *
                ;* VALUE FOR EACH RESPECTIVE PORT HAS EXPIRED.  ALL FOUR ROUTINES    *
                ;* MUST BE SERVICED BETWEEN INPUT CAPTURES THAT OCCUR WITHIN ZERO    *
                ;* CROSSINGS OF 60 Hz AC, THAT IS, EVERY 8.33 mS.                     *
                ;*********************************************************************

 B760 8601      OUTSERV1: LDAA    #OUT1             ;OUTPUT 1
 B762 A704                STAA    PORTB,X           ;TURN IT ON
 B764 B6B7A4              LDAA    PULWID            ;FOR ABOUT 20 uS
 B767 4A        WAIT1:    DECA
 B768 26FD                BNE     WAIT1             ;WAIT FOR TRIAC TO TRIGGER
 B76A 1D0401              BCLR    PORTB,X,OUT1CLR   ;TURN IT OFF
 B76D 1D23BF              BCLR    TFLG1,X,FLG1CLR   ;CLEAR OC2F FLAG
 B770 3B                  RTI
 B771 8602      OUTSERV2: LDAA    #OUT2             ;OUTPUT 2
 B773 A704                STAA    PORTB,X
 B775 B6B7A4              LDAA    PULWID
 B778 4A        WAIT2:    DECA
 B779 26FD                BNE     WAIT2
 B77B 1D0402              BCLR    PORTB,X,OUT2CLR
 B77E 1D23DF              BCLR    TFLG1,X,FLG2CLR   ;CLEAR OC3F FLAG
 B781 3B                  RTI
 B782 8604      OUTSERV3: LDAA    #OUT3             ;OUTPUT 3
 B784 A704                STAA    PORTB,X
 B786 B6B7A4              LDAA    PULWID
 B789 4A        WAIT3:    DECA
 B78A 26FD                BNE     WAIT3
 B78C 1D0404              BCLR    PORTB,X,OUT3CLR
 B78F 1D23EF              BCLR    TFLG1,X,FLG3CLR   ;CLEAR OC4F FLAG
 B792 3B                  RTI
```

```
 B793 8608      OUTSERV4: LDAA    #OUT4             ;OUTPUT 4
 B795 A704                STAA    PORTB,X
 B797 B6B7A4              LDAA    PULWID
 B79A 4A        WAIT4:    DECA
 B79B 26FD                BNE     WAIT4
 B79D 1D0408              BCLR    PORTB,X,OUT4CLR   ;TURN IT OFF
 B7A0 1D23F7              BCLR    TFLG1,X,FLG4CLR   ;CLEAR OC5F FLAG
 B7A3 3B                  RTI
 B7A4                     page


** FADER  Version 1.50 **    PAGE 7    Sun Oct 11 14:40:09 1992

                ;*****************************************************************
                ;*                         CONSTANTS AND VARIABLES                *
                ;*****************************************************************

 B7A4 10        PULWID:  DFB     010H
 B7A5 3000      MAX:     DWM     3000H        ;DIMMEST-150 DELAY/30 COND
 B7A7 0800      MIN:     DWM     0800H        ;BRIGHTEST-30 DELAY/150 COND
 B7A9 00        DECFLAG: DFB     00H
 B7AA FF        INCFLAG: DFB     0FFH
 B7AB 3000      INIT1:   DWM     3000H        ;DIMMEST PORT 1
 B7AD 1900      INIT2:   DWM     1900H        ;BRIGHT GOING DIM PORT 2
 B7AF 1900      INIT3:   DWM     1900H        ;DIM GOING BRIGHT PORT 3
 B7B1 0800      INIT4:   DWM     0800H        ;BRIGHTEST PORT 4
 0100                    ORG     $0100
 0100           DELAY1:  DFS     2
 0102           DELAY2:  DFS     2
 0104           DELAY3:  DFS     2
 0106           DELAY4:  DFS     2
 0108           FLAG1:   DFS     1
 0109           FLAG2:   DFS     1
 010A           FLAG3:   DFS     1
 010B           FLAG4:   DFS     1
 0000                    END


** FADER  Version 1.50 **    PAGE 8    Sun Oct 11 14:40:09 1992

0030  ADCTL              B670  CLEAR              0079  CLRFLAG
00FE  CLRIC3             0021  CONVERT            0001  CROSS
```

```
B7A9  DECFLAG          0100  DELAY1          0102  DELAY2
0104  DELAY3           0106  DELAY4          B6A0  DOWN1
B6D2  DOWN2            B704  DOWN3           B736  DOWN4
0003  EDGEDET          0080  ENABA2D         0032  FADRATE
0108  FLAG1            0109  FLAG2           010A  FLAG3
010B  FLAG4            00BF  FLG1CLR         00DF  FLG2CLR
00EF  FLG3CLR          00F7  FLG4CLR         B600  GENINIT
B7AA  INCFLAG          B7AB  INIT1           B7AD  INIT2
B7AF  INIT3            B7B1  INIT4           B677  INPUT
0078  INTMASK          007E  JUMPCOD         B66B  LODRATE
B7A5  MAX              B7A7  MIN             0039  OPTION
0001  OUT1             0001  OUT1CLR         0002  OUT2
0002  OUT2CLR          0004  OUT3            0004  OUT3CLR
0008  OUT4             0008  OUT4CLR         B760  OUTSERV1
B771  OUTSERV2         B782  OUTSERV3        B793  OUTSERV4
B673  POLL             0004  PORTB           B7A4  PULWID
00DC  PVOC2            00D9  PVOC3           00D6  PVOC4
00D3  PVOC5            1000  REGBAS          00FF  SETOUT
01FF  STACK            B66F  START           B693  START1
B6C5  START2           B6F7  START3          B729  START4
000E  TCNT             0020  TCTL1           0021  TCTL2
0023  TFLG1            0014  TIC3            0022  TMSK1
0018  TOC2             001A  TOC3            001C  TOC4
001E  TOC5             B6B1  TOGGLE1         B6E3  TOGGLE2
B715  TOGGLE3          B748  TOGGLE4         B6B6  UP1
B6E8  UP2              B71A  UP3             B74E  UP4
B767  WAIT1            B778  WAIT2           B789  WAIT3
B79A  WAIT4
```